

DLite Script Manual

Contents

Chapter 1 Using DLite Script	3
1-1 WHAT IS DLITE SCRIPT?	3
1-2 HOW DLITE SCRIPT WORKS?	3
1-3 BASIC SCRIPT SYNTAX	4
Chapter 2 DLite script editor	10
2-1 DLITE SCRIPT EDITOR INSTALLATION	10
2-2 EDIT YOUR FIRST SCRIPT BY SCRIPT EDITOR	12
2-3 BMP MESSAGE	17
2-4 ANIMATION MESSAGE	18
Chapter 3 Script example	19
3-1 FIRST TEXT EXAMPLE	19
3-2 SAMPLE FOR STANDARD TEXT	20
3-3 SAMPLE FOR SMALL FONT (5x7)	21
3-4 BMP MESSAGE EXAMPLE	21
3-5 BMP MIXED WITH TEXT MESSAGE	22
3-6 SCRIPT ANIMATION	23
3-7 MORE INTERESTING EFFECTS BY SCRIPT	25
3-8 UPLOAD SCRIPT FILES TO LED DEVICE	27
About eSoftsystem Corp.	28

Chapter 1 Using DLite Script


1-1 What is DLite script?

DLite script is a command set for display messages inside the DLite LED. It's a "plain text" based file that with specific file name "ekan.ils". If you want to add complex effects to your message, the best way is using DLite script to archive those effects. You can mix graphic and text message in the same display, making the animation effects. DLite script gives you specific ability to edit the message file.

1-2 How DLite script works?

The DLite firmware will parse the DLite script when LED power up, or after you upload the new DLite script file from message editor/script editor.

When parsing the script file, DLite firmware will also look up the BMP files that used by "ekan.ils" script file. If all process normally, the firmware will start the message display. The whole process usually takes less than 1 second; normally user won't feel this parsing process.



```
ekan.ils - 記事本
檔案(E) 編輯(E) 格式(O) 說明(H)

CLEAR
IN GO-UP 1 50
TEXT 0 0 ORANGE BLACK " DLite 1.5 "
DELAY 1000
OUT GO-UP 1 50

CLEAR
IN GO-UP 1 50
TEXT 0 0 RED BLACK "Demo for QC"
DELAY 3000
OUT GO-UP 1 50

CLEAR
IN GO-DOWN 1 100
IMG 0 0 GREEN BLACK 1.bmp
TEXTS 34 0 ORANGE BLACK "DLite 1.5"
TEXTS 34 9 ORANGE BLACK "Ver 1.5.57"
DELAY 3000
OUT GO-DOWN 1 100

CLEAR
IN GO-DOWN 1 100
IMG 0 0 GREEN BLACK 2.bmp
TEXTS 38 9 ORANGE BLACK "250"
TEXTS 67 9 ORANGE BLACK "1200"
```

1-3 Basic script syntax

There are 3 different kinds of command set. List on following table:

- Screen swap commands
- Effects
- Message
- Graphic function

Here is the complete list:

A: Screen swap commands

Type	Function name	Parameter	Explanations
Screen swap command	IN	IN GO-LEFT point millisecond	IN GO-LEFT 1,100 Which means, the following text will "ENTER" LED screen from right side to left side; every 100 millisecond with 1 point movement until the end of text.
	OUT	OUT GO-LEFT point millisecond	OUT GO-LEFT 1,100 Which means, the following text will "LEAVE" LED screen to left side; every 100 millisecond with 1 point movement until the end of text.

B: Effects

Type	Function name	Parameter	Explanations
Effects	GO-LEFT	GO-LEFT point, millisecond	
	GO-RIGHT	GO-RIGHT point millisecond	
	GO-UP	GO-UP point millisecond	
	GO-DOWN	GO-DOWN point millisecond	
	FADE	FADE millisecond	The text or graph on screen will fade out on designate time interval, total fade out time will be 4 x milliseconds.
	DELAY	DELAY millisecond	DELAY 1000 means the screen will delay for 1000 millisecond
	BLINK	BLINK times millisecond	BLINK 5 100 means BLINK 5 times and each time delay 100 millisecond

C: Message

Type	Function name	Parameter	Explanations
Message	TEXT	TEXT PosX PosY ForeColor BackColor "TEXTSTRING"	Normal ASCII text with font size 8x16, the Position 0, 0 means upper left corner, and the 96, 16 means bottom right corner of the 1 st LED display screen. And the forecolor and backcolor belongs to 1 of the 4 color: (GREEN RED ORANGE BLACK) For Example: TEXT 0 0 GREEN BLACK "Hello World!" will show the "Hello World!" with Green color text on the LED display.
	TEXTS	TEXTS PosX PosY ForeColor BackColor "TEXTSTRING"	Small ASCII text with font size 8x6, the Position 0, 0 means upper left corner, and the 96, 16 means bottom right corner of the 1 st LED display screen. And the forecolor and backcolor belongs to 1 of the 4 color: (GREEN RED ORANGE BLACK) For Example: TEXTS 0 0 GREEN BLACK "Hello World!" will show the small font of "Hello World!" with Green color text on the LED display.
	TEXTB5	TEXTB5 PosX PosY ForeColor BackColor "Chinese TEXTSTRING"	Tradition Chinese text with font size 16x16,

	IMG	IMG PosX, PosY, ForeColor,Back Color, "Filename.bmp"	<p>This command will display the designate BMP file. Position 0, 0 means upper left corner, and the 96, 16 means bottom right corner of the 1st LED display screen. And the forecolor and backcolor belongs to 1 of the 4 color:</p> <p>(GREEN RED ORANGE BLACK)</p> <p>For Example:</p> <p>IMG 0 0 GREEN BLACK "test.bmp" will show the "test.bmp" with Green color on the LED display.</p>
	DATE	DATE PosX, PosY, ForeColor, BackColor, Second	<p>This command will display the DATE variable. Position 0, 0 means upper left corner, and the 96, 16 means bottom right corner of the 1st LED display screen. And the forecolor and backcolor belongs to 1 of the 4 color: (GREEN RED ORANGE BLACK)</p> <p>And the last parameter is the seconds that DATE variable stay at screen.</p> <p>For Example:</p> <p>DATE 0 0 GREEN BLACK 3 will show DATE variable 3 seconds with Green color on the LED display.</p>

	<p>TIME</p>	<p>TIME PosX PosY ForeColor BackColor Second</p>	<p>This command will display the TIME variable. Position 0, 0 means upper left corner, and the 96, 16 means bottom right corner of the 1st LED display screen. And the forecolor and backcolor belongs to 1 of the 4 color: (GREEN RED ORANGE BLACK)</p> <p>And the last parameter is the seconds that TIME variable stay at screen.</p> <p>For Example:</p> <p>TIME 0 0 GREEN BLACK 3 will show the TIME variable 3 seconds with Green color on the LED display.</p>
--	-------------	---	--

D: Graphic

Type	Function name	Parameter	Explanations
Graphic Function	PIXEL	PIXEL PosX PosY ForeColor BackColor	Draw a pixel on LED display
	LINE	LINE PosX1, PosY1, PosX2, PosY2, Color	Draw a line form X1, Y1 to X2, Y2, and the color belongs to 1 of the 4 color: (GREEN RED ORANGE BLACK)
	CLRECT	CLRECT PosX1, PosY1, PosX2, PosY2	CLEAR the rectangle area of X1, Y1, X2, Y2
	FILLRECT	FILLRECT PosX1, PosY1, PosX2, PosY2, Color	FILL the rectangle with specific color
	FILL	FILL color	FILL the screen with specific color. And the color belongs to 1 of the 4 color: (GREEN RED ORANGE BLACK)
	CLEAR		CLEAR the screen and back to black g
Comment	#	#Comment Use, will ignore the this line of script	

Chapter 2 DLite script editor


DLite script editor is a text based editor; help you to generate script more quickly and more correctly. It has 3 main functions:

- **Text based editing environment**
- **Script preview emulator**
- **Upload and download interface**

Following is the script editor introduction for script editor.

2-1 DLite script editor installation

DLite script editor was put in the CD-ROM's DLite script folder; you can double click the setup icon to start the set up process.



Step 1: Click to DLiteScriptSetup icon

Step 2: Click the icon to start the installation process.




Have installation on Windows XP SP2

The DLite script editor may have security conflicts with some Windows XP SP2 configuration. If installation problem happened, please copy the installation folder to your desktop then re-start the installation process.

After successful installation, you can start this program from the **[start]** menu, **[eSoftsystem]** then **[DLite script editor]** to start the program.

The DLite script editor's interface looks like following figure:




Here is step by step sample guide for how to use script editor:


2-2 Edit your first script by script editor

DLite script editor have build-in some common script generation page, user could use those tools, quickly generate scripts for later modify.


Please click the **[TEXT]** button on the toolbar menu.



Click on the tools bar [Text] button




Shows script generate window for the editor, Please input "Welcome!" on the text input box. Select the effect, color, font you want, Then Click **[Apply]** button.




The script will be generated into editing windows

Now, we already have the first message, next step is preview the script. Please press **[Preview]** button on the toolbar for preview the message.




Press the **[Preview]** button, DLite script editor will ask you to save the file first, the preview script file name must be ekan.ils



You can adjust the preview module and cell size, to reflect actual LED module setting, Press [**Emulator Start**] button to preview system setting


If the script previews result is correct, you can upload the script to the LED display. Please press upload button to upload the script.



Press [**Upload**] button to open upload dialog box

Then input the target IP address, input the password, default is "admin", clean LED display flash memory if needed. Please select the ekan.ils file just saved

Add all files needed into upload list, then press **[Upload selected file]** button.



If upload succeed, will show upload success message. You can see the message on the LED display as you saw in the emulator before.



If upload file fail

The default speed for animation message is 100 ms moves 4 pixels, and the BMP size is 16x16 pixels. Please keep the refresh speed about 100 ms to get property animation effects. Please make sure your IP address and password both are correct, and the file format is correct. DLite only accepts text file ekan.ils and 1-bit BMP.




Please remember to clean LED storage regularly

Because DLite only have about 300 K bytes memory for script and BMP file, so please remember delete unnecessary file before you upload file to LED display.

2-3 BMP message


DLite support BMP message, you can add your company's logo, or any kind of graphic message into LED display.

Edit BMP message almost like text based message, the only different is BMP message is using file name instead of plain text. You can create the BMP by using Microsoft Paint or other 3rd party graphic tool.



Click the BMP message icon


Select the BMP file, the file format must be 1-bit BMP (black and white), then select the effect for the BMP message.



You can preview the BMP message on the preview window, just like the text message in previous section.


2-4 Animation message

DLite script editor could help you to generate animation script. You can load 4 BMP files, **each BMP files size is 16x16 pixels** to create simple animation effects. Following is steps for creating animation scripts:




Press animation button

Load 4 BMP file to the animation message generator, the BMP format must be 16x16 pixels with 1-bit color.



Press **[Preview]** and **[Start]** button to see the animation effects, Press **[Return Main]** to back to script editor.



Preview the effects, Then you can modify the script for more interesting effects.

Chapter 3 Script example

3-1 First text example

Before you start, there is some bmp sample file located under your application directory/SampleBMP folder, and some script files comes with the Visual Basic sample under SDK folder.

Here is the first example, illustrates basic message technique: display message, delay for 3 seconds, clear screen then display next message.

```
#Script Start
TEXT 0 0 GREEN BLACK "Hello World!"
DELAY 3000
CLEAR
TEXT 0 0 RED BLACK "1st Message"
DELAY 3000
CLEAR
TEXT 0 0 ORANGE BLACK " Welcome! "
DELAY 3000
CLEAR
#Script End
```



The important "CLEAR" command

Please use **CLEAR** after display message to make sure that new message won't overlap on old message, avoid illusion of image corruption.



If NO script on LED display?

LED will display the system default message, like IP address/Date/Time...etc. **BUT** if you send a Instant Message on Web Messenger or using API to send instant message without clean the screen. The EKAN will stay at last message that displayed.

3-2 Sample for Standard Text

Here is complete sample for text message, following is a text scroll in from the left side of led display, than Delay 1 second, then scroll-up out of screen.

```
#Script Start  
  
IN GO-LEFT 1 50  
TEXT 0 0 GREEN BLACK "Text Message"  
DELAY 1000  
OUT GO-LEFT 1 50  
CLEAR  
  
#Script End
```



When DLite parsing error on script

DLite won't crash because bad script. Usually DLite will skip the line that has mistake or bad formatted, but if DLite won't find bmp files that designated on the script, It will show what BMP file is missed in the LED file system.

3-3 Sample for small font (5x7)

DLite are built-in with 5x7 small font support, so the single LED display can play 2 lines of small font message.

```
#Script Start

TEXTS 0 0 GREEN BLACK "TEST UPPER TEXT"
DELAY 2000
TEXTS 0 9 RED BLACK "TEST BOTTOM TEXT"
DELAY 2000
OUT GO-DOWN 1 100
CLEAR

#Script End
```

3-4 BMP message example

Here is the sample that displays a BMP message every 100 milliseconds move 1 point, delay 5 seconds, and then exits the screen.

```
#Script Start

IN GO-DOWN 1 100
IMG 0 0 GREEN BLACK 1.bmp
DELAY 5000
OUT GO-DOWN 1 100
CLEAR

#Script End
```

Of course, you could also cancel IN and out effects for the BMP message, the sample will become:

```
#Script Start

IMG 0 0 GREEN BLACK 1.bmp
DELAY 5000
CLEAR

#Script End
```

3-5 BMP mixed with text message

BMP message can mix with text message at the same time; create more interest or more attractive message. Here is the typical use of mix BMP with text message.

The following sample will load the 1.bmp and scroll down, then add 2 small font texts on the BMP. The **BMP will act as message background template**. And the overlapped text could as variable on the message template.

```
#Script Start

IN GO-DOWN 1 100
IMG 0 0 GREEN BLACK 1.bmp
TEXTS 34 0 ORANGE BLACK "DLite 1.5"
TEXTS 34 9 ORANGE BLACK "Ver 1.5.57"
DELAY 3000
OUT GO-DOWN 1 100
CLEAR

#Script End
```

3-6 Script animation

□ **The basic theory of animation**

Animation is using the persistence of vision with regard to changing images. For example, if you stare at some images and images switch faster than 1/16 seconds, then the eyes will take those images as one animated picture. Normal animation need 20+ pictures per second, but we could use much more less pictures on LED display.

□ **How to use script to make animation**

Basically, we just need to copy and paste the script block, and modify display bmp file and moving the position, then can create animation effects. Here are the examples:

□ **Static type animation**

You can create static animation by switch static graphics in the same position. Following is the script for the static position animation, will show the rotating sandglass at the left side of the LED display. Here is static text effect the swap the text color every second. You could find the sample under ANI_COLORSWITCH folder

```
#Script Start

TEXT 0 0 ORANGE BLACK "D"
TEXT 8 0 RED BLACK "E"
TEXT 16 0 ORANGE BLACK "M"
TEXT 24 0 GREEN BLACK "O"
DELAY 2000
TEXT 0 0 GREEN BLACK "D"
TEXT 8 0 ORANGE BLACK "E"
TEXT 16 0 RED BLACK "M"
TEXT 24 0 ORANGE BLACK "O"
DELAY 2000
  CLEAR
  DELAY 1000
```

#Script End

□ **Moving animation**

Moving animation usually combined with switch position and change display bmp. Of course, you could mix with change color at the same time. But you might need to manual edit the script to get the best effect.

PAC1P to PAC4P are PACMAN bmp, you could found it at your SampleBMP folder that located under your program installation folder.

```
#Script Start  
  
IMG 0 0 ORANGE BLACK PAC4P.BMP  
DELAY 200  
IMG 4 0 GREEN BLACK PAC3P.BMP  
DELAY 200  
IMG 8 0 RED BLACK PAC2P.BMP  
DELAY 200  
IMG 12 0 ORANGE BLACK PAC1P.BMP  
DELAY 200  
IMG 16 0 GREEN BLACK PAC4P.BMP  
DELAY 200  
IMG 20 0 RED BLACK PAC3P.BMP  
DELAY 200  
IMG 24 0 ORANGE BLACK PAC2P.BMP  
DELAY 200  
IMG 28 0 GREEN BLACK PAC1P.BMP  
DELAY 200  
CLEAR  
  
#Script End
```

3-7 More interesting effects by script

Following is more interesting script for you, it mixed all technique discussed in this chapter. And add some tricky effect for you. You could get the sample script.

```
#Script Start

TEXT 48 0 ORANGE BLACK "W"
DELAY 500
TEXT 56 0 RED BLACK "i"
DELAY 500
TEXT 64 0 GREEN BLACK "n"
DELAY 500
TEXT 72 0 RED BLACK "C"
DELAY 500
TEXT 80 0 ORANGE BLACK "o"
DELAY 500
TEXT 88 0 GREEN BLACK "n"
DELAY 500
TEXT 96 0 ORANGE BLACK "-"
DELAY 500
TEXT 104 0 GREEN BLACK "8"
DELAY 500
TEXT 112 0 RED BLACK "0"
DELAY 500
TEXT 120 0 ORANGE BLACK "0"
DELAY 500
TEXT 128 0 GREEN BLACK "0"
DELAY 500
TEXT 136 0 RED BLACK " "
TEXT 144 0 RED BLACK " "
DELAY 2000
CLEAR

TEXT 0 0 RED BLACK "P"
TEXT 8 0 GREEN BLACK "rogrammable"
DELAY 1000
TEXT 104 0 RED BLACK "A"
TEXT 112 0 GREEN BLACK "utomation"
DELAY 1000
OUT FADE 500
CLEAR
```

```
TEXT 0 0 RED BLACK "C"
TEXT 8 0 GREEN BLACK "ontroller"
TEXT 80 0 ORANGE BLACK " "
TEXT 88 0 ORANGE BLACK " "
TEXT 96 0 ORANGE BLACK " "
TEXT 104 0 GREEN BLACK "WinCon-8000"
DELAY 1000
TEXT 104 0 RED BLACK "WinCon-8000"
DELAY 500
TEXT 104 0 ORANGE BLACK "WinCon-8000"
DELAY 500
TEXT 104 0 GREEN BLACK "WinCon-8000"
DELAY 1000
OUT FADE 500
CLEAR

IN GO-DOWN 1 30
TEXT 0 0 ORANGE BLACK "Integration with Windows"
DELAY 1000
OUT GO-DOWN 1 30
CLEAR

TEXT 24 0 RED BLACK "CE.NET"
DELAY 500
TEXT 72 0 GREEN BLACK " & "
DELAY 500
TEXT 96 0 RED BLACK "RISC CPU"
DELAY 500
OUT GO-UP
CLEAR
IN GO-UP 1 30
TEXT 48 0 ORANGE BLACK "Combine with"
DELAY 1000
OUT GO-UP 1 30
CLEAR

TEXT 56 0 RED BLACK "IPC"
DELAY 500
TEXT 80 0 GREEN BLACK " & "
DELAY 500
TEXT 104 0 RED BLACK "PLC"
DELAY 500
OUT GO-UP
CLEAR
```

```
IN GO-LEFT 1 20
TEXT 0 0 GREEN BLACK "Industrial PC standard peripherals"
DELAY 1000
OUT GO-LEFT 1 20
CLEAR
#Script End
```

3-8 Upload script files to LED device

After you finished the script editing, you could use the following 2 way to upload your data.

□ **Using Ethernet:**

You can use the script editor program or xDLite.ocx with the upload method to upload the data, or using the script editor to upload download data.

□ **Using RS-232:**

You can use 7188xw program come with the CD-ROM or the ICPDAS MiniOS tools to upload the data. But, you have to manually switch the LED device into init mode. You could refer firmware update method on the Quickstart manual



Don't always upload, remember to delete unnecessary file

DLite won't automatically delete previous file until you using command to do so, so please delete unnecessary file before you upload new script.

About eSoftsystem Corp.

eSoftsystem Technology Corp. is the most innovative embedded solution provider. It has built up a team of world experts in embedded software and hardware systems to provide customer high-performance and high-quality embedded solution product and services.